
1 Vorwort 

1.1 Idee 
Für den NDR-Klein-Computer (NKC) wurden schon viele Platinen von Enthusiasten - lange nach der 
eigentlichen Ära des NKCs - entwickelt. Dabei ging es aber oft um Speicher- oder allgemeine I/O-
Projekte. 

Dies nahm ich zum Anlass, mal einen Exoten zu entwickeln - nämlich den Einstieg des NKCs in die 
mediale Welt mit einer Tunerkarte …  

1.2 Ansatz 
Es sollte eine Tunerkarte entwickelt werden, die sich mit vertretbarem Hardwareaufwand aufbauen 
und auch recht einfach in Assembler programmieren ließ, ohne dabei der Oberexperte zu sein. 

Deshalb fiel die Wahl auf ein Konzept, das den I2C-Bus von Philips nutzt, der für verschiedenste 
Bausteine z.B. innerhalb eines Fernsehgerätes für die Programmwahl und diverse Einstellungen wie 
Helligkeit, Kontrast, Lautstärke und Klang genutzt wird.  

1.3 Erweiterungen 
In Zukunft sollte die Tunerkarte noch einen RDS-Dekoder für das Radio bekommen. Dies war 
zunächst eine Standalone-Karte mit einem ATMEGA168, die Huckepack auf die Tunerkarte gesetzt 
wurde. Die dekodierten RDS-Zeichen sollen von dem ATMEGA168 per serielle Schnittstelle an den 
NKC übergeben werden. Des Weiteren sollte die Regelspannung des Tuners (AGC) zur Anzeige der 
Signalstärke eines Senders mit Hilfe einer AD-Wandler-Karte verwendet werden. 

1.4 Aktueller Stand 
Thomas Heller und Matthias Kübeck haben nun eine industrielle Platine mit den entsprechenden 
Dateien zum PCB-Layout erstellt. Es gibt Prototypplatinen, die fehlerfrei funktionieren. 

Die Platine enthält die Logik für den I2C-Bus, den RDS-Dekoder und die Aufbereitung des Tuner-
AGC-Signals zur Auswertung der Signalstärke eines Senders. 

 

2 Beschreibung des Konzepts  
Das Konzept der Tuner-Karte besteht im Wesentlichen darin,  ein Interface des NKCs zum Philips 
I2C-Bus bereitzustellen. Der I2C-Bus ist ein serieller Bus mit einer Taktleitung (SCL) und einer 
Datenleitung (SDA). Genaueres ist dem Dokument „UM10204.pdf - I2C-bus specification and user 
manual“ zu entnehmen.  

Es gibt Master- und Slave-Geräte auf diesem Bus. Ein Master-Gerät steuert den Datenfluss beim 
Lesen und Schreiben gegenüber einem Slave-Gerät. Ein Master ist i.d.R. ein Microcontroller. 
Grundsätzlich lässt der I2C-Bus mehrere Master-Geräte und natürlich mehrere Slave-Geräte zu.  

Im vorliegenden Fall der NKC-Tuner-Karte haben wir folgende Konstellation für den I2C-Bus:   

 

Gerät Funktion Baustein Beschreibung Datenblatt 

Master I2C-Prozessor PCF 8584-P Der Baustein stellt das Interface 

zwischen dem parallelen NKC-

Bus und dem seriellen I2C-Bus 

dar 

PCF8584.pdf 

Slave Tuner Philips FM1216 

ME / I H-3 

I2C-programmierbarer 

Tunerbaustein für Radio und 

analoges Fernsehen 

FM1216ME_ 

MK3.pdf 

Slave Sound TDA 8425 Lautstärke- und Klangeinstel-lung 

mit zwei Eingangskanä-len 

(Tunerkarte liegt auf erstem 

Kanal!) 

TDA8425.pdf 

Slave Speicher –  

2 KByte 

AT24C02 EEPROM zum residenten Spe-

ichern der aktuellen Settings des 

Tuners und des Soundbausteins 

für Neustart sowie der Programme 

(Frequenzen) für Radio (und 

Fernsehen) 

AT24C02.pdf 

Slave RDS-Emfang / 

Verarbeitung 

ATMEGA 168-P Microcontroller zur Aufbereitung 

des RDS-Datenstroms sowie der 

Signalstärke und Übermittlung an 

den NKC 

ATMEGA168.pdf 

3 Schaltungsprinzip 
 

  

3.1 Adresslogik  
Die Adresslogik wird mit dem Baustein 74HCT688 abgebildet. Er benötigt zur Freigabe das IO-
Request-Signal (IORQ)  des NKC-Busses. Des Weiteren werden die Datenleitungen A1 bis A7 
zugeführt und mit den Werten, die am JP1 (siehe Schaltplan) eingestellt werden, verglichen. Am JP1 
wird somit die IO-Adresse des I2C-Prozessors PCF 85 84-P eingestellt . Bemerkung: Da der 
74LS688 8 Zustände vergleicht, wir aber nur 7 benötigen, wurden die Eingänge P0 und Q0 (siehe 
Schaltbild 74HCT688) auf 0V gelegt, damit diese immer gleich sind. Die Datenleitung A0  ist direkt an 
den I2C-Prozessor PCF 8584-P  geführt, um dort interne Register anzusprechen. Sind die angelegten 
Adressdaten des JP1 und der Datenleitungen A1 bis A7  identisch, dann wird das Signal –Chip 
Select  ausgelöst, was einerseits den Datenpuffer 74HCT245  und den I2C-Prozessor PCF 8584-P  
frei gibt. 

3.2 Datenpuffer 
Als Datenpuffer wird ein 74HCT245  eingesetzt. Seine Freigabe erfolgt durch das Signal –Chip 
Select . Die Richtung, also ob Daten zum I2C-Prozessor PCF8584  gesendet oder von ihm geholt 
werden, bestimmt das Signal –RD vom NKC-Bus. Die zu sendenden oder zu holenden Daten werden 
über die Datenleitungen D0 bis D7  vom NKC-Bus  übertragen. 

3.3 I2C-Prozessor 
Der I2C-Prozessor PCF 8584-P  wird durch das Signal –Chip Select  freigegeben. Mit der 
Adressleitung A0 vom NKC-Bus  wird entschieden, ob das Datenregister  (Zustand = logisch 1) oder 
das Status-register  (Zustand = logisch 0) angesprochen wird. Daten, um aus dem I2C-Prozessor 
PCF 8584-P zu lesen oder auf ihn zu schreiben, gelangen über die durch den 74HCT245 gepufferten 
Datenleitungen D0 bis D7  zu ihm. Der Zugriff des Lesens wird über das Signal –RD und der des 
Schreibens über das Signal –WR des NKC-Busses ausgelöst. Für eine korrekte Funktion benötigt der 
I2C-Prozessor PCF 8584-P  noch den CPU-Takt vom NKC-Bus. Zudem wurde das –Reset-Signal  
des NKC-Busses an den Prozessor gelegt, um ihn bei einem System-Reset zurückzusetzen. 

3.4 I2C-Bus 
Der I2C-Prozessor PCF 8584-P  übersetzt die parallelen 8-Bit-IO-Daten  in die seriellen Daten des 
I2C-Busses  um. Dieser Bus besitzt eine Taktleitung (SCL)  und eine Datenleitung (SDA).  An diesen 
beiden Leitungen hängen alle Geräte des I2C-Busses . In unserem Fall (neben dem I2C-Prozessor) 
der Tuner, der Soundbaustein, der Speicher - das EEPROM und der ATMEGA 168-P (siehe Tabelle 
oben). 

3.4.1 I2C-Bus-Adressierung 
Ähnlich wie bei den IO-Adressen auf dem NKC-Bus  müssen alle Geräte auf dem I2C-Bus  
eine Adresse (I2C module adress MAD)  besitzen, über die sie angesprochen werden. Wenn 
Geräte einen Schreib- und einen Lesezugriff zulassen, dann besitzen sie für jeden dieser 
Zugriffe eine eigene I2C-Adresse . Die folgende Tabelle zeigt die verwendeten I2C-Adressen : 

 

Gerät Funktion Baustein I2C-Adresse (MAD) Bemerkung 

Master I2C-Prozessor PCF 8584-P $AA Adresse wird im Bedienpro-

gramm bei der Initialisierung 

mitgegeben. Baustein wird nur im 

Master-Betrieb eingesetzt. Adres-

se ist nicht weiter relevant, muss 

aber gesetzt werden! 

Slave Tuner_HF Philips FM1216 

ME / I H-3 

$C0 - Schreibzugriff 

$C1 - Lesezugriff 

Adresse ist fest, kann aber in be-

stimmten Grenzen durch Beschal-

tung von PIN 6 geändert werden 

(siehe Datenblatt FM1216ME_ 

MK3.pdf). In der Schaltung liegt 

PIN 6 auf 0V. 

Slave Tuner_IF Philips FM1216 

ME / I H-3 

$86 - Schreibzugriff 

$87 - Lesezugriff 

Adresse ist fest, nicht änderbar. 

Slave Sound TDA 8425 $82 - Schreibzugriff Adresse ist fest, nicht änderbar. 

Slave Speicher –  

2 KByte 

AT24C02 $A0 - Schreibzugriff 

$A1 - Lesezugriff 

Adresse ist fest, kann aber in be-

stimmten Grenzen durch Beschal-

tung der PINs A0 bis geändert 

werden (siehe Datenblatt 

AT24C02.pdf). In der Schaltung 

liegen alle PINs auf 0V. 

Slave RDS-Emfang /  

Signalstärke 

ATMEGA 168-P $?? - Schreibzugriff 

$?? - Lesezugriff 

Erst in der Nächsten 

Version der Spezifikation! 

4 Schaltplan 

5 Anmerkungen 

5.1 Spannungsversorgung: 
Um HF-Störungen von den Spannungsversorgungen fernzuhalten, wurden folgende Kombinationen 
aus Spule und Kondensator für die verschiedenen Bausteine eingesetzt: 

 

Baustein Spule Kondensator 

Philips FM1216 ME / I H-3 - HF 10 µH 10 nF 

Philips FM1216 ME / I H-3 - IF 100 µH 10 nF 

TDA 8524 10 µH 10 nF 

TDA 7330 100 µH 100 nF 

UA 741-P -- -- -- 10 nF 

ATMEGA 168-P 100 µH 100 nF 

 

5.2 Eingänge und Ausgänge: 
Hier werden alle Ein- und Ausgänge sowie Steckverbinder und Jumper der Tunerkarte beschrieben: 

 

Steckerleiste/ 

Buchse 

Beschreibung Standard Bemerkung 

JP1 Jumper zur Einstellung 

der NKC-IO-Adresse 

des I2C-Prozessors 

PCF 8584-P bzw. der 

Tuner-Karte 

ja Im Bedienprogramm ist die  IO-Adresse 

$FFFFFF52 verwendet. 

JP3 Zweite Sound-IF bei TV-

Empfang 

optional Mit einem weiteren Baustein wird 2-Kanal- 

oder Stereo-TV-Empfang ermöglicht. 

SV1 ISP-Programmier-

anschluss (SPI) 

ATMEGA 168-P 

ja 1 = GND, 2 = MOSI, 3 = VTG, 4 = RESET, 5 

= SCK, 6 = MISO 

SV2 Serielle Schnitt-stelle für 

den RDS-Betrieb 

ja Übertragung von RDS-Zeichen an den NKC 

und Steuerung des RDS-Dekoders                 

1 = TX, 2 = GND, 3 = RX 

 

SV3 AGC-Regelspan-nung 

des Tuners zur Anzeige 

der Signalstärke eines 

Senders 

ja 1 = out OP-Verstärker 741, 2 = GND, 3 = out 

- einstellbar 

SV4 Eingang 2. Kanal des 

Sound-bau-steins TDA 

8425 

optional Man kann hier z.B. den Ausgang der NKC-

Soundplatine anschlie-ßen. Auf Kanal 1 liegt 

der Tuner! 

SV5 I2C-Bus optional Hier können weitere Bausteine an den I2C-

Bus angeschlossen werden. 

X1 Audio-Signal - rechts ja Anschluss eines NF-Verstärkers 

X2 Audio-Signal - links ja Anschluss eines NF-Verstärkers 

X3 Video-Signal (TV) ja CVBS- oder BAS-Signal an 75 Ohm 

Impedanz 

Antennenein-

gänge Radio 

und TV 

HF-Eingang mit 75 Ohm 

Impedanz 

ja Wichtig:  Der NKC „sendet“ mit seinem CPU-

Takt und dem Takt der Grafikkarte in der Nä-

he des UKW-Bandes oder zumindest dessen 

IF von 10,7 MHz. Deshalb sollte man an die 

Antennenbuchse auch eine etwas abseits 

stehende Antenne mit Koaxialkabel und –

stecker anschließen. Sonst rauscht es even-

tuell nur, obwohl die Karte funktioniert und 

richtig programmiert ist! Ein UKW-Bandpass  

(87,5 -108 MHz) vor dem Radio-Antennen-

eingang hilft ebenfalls, Störungen zu vermei-

den und erhöht die Selektivität des Tuners. 

6 Test und Programmierung der Karte 
Die Tunerkarte kann mit den im Folgenden beschriebenen Programmteilen - am Beispiel von 680XX-
Assembler getestet werden. Es existiert dazu die Datei ‚tunertst.asm‘ . 

6.1 Programmierung mit Schichtenkonzept: 
Die Programmierung der Tuner-Karte erfolgt in einem Schichtenkonzept : 

In der ersten Schicht  gibt es Basisroutinen, um den I2C-Prozessor  auf seinem Statusregister  oder 
seinem Datenregister  anzusprechen – mit schreibendem oder lesendem Zugriff . 

In der zweiten Schicht  gibt es Basisroutinen, um den I2C-Prozessor zu initialisieren, n Daten-
bytes an den I2C-Bus zu senden oder von ihm zu lesen. 

In der dritten Schicht  gibt es Routinen , die es erlauben, die einzelnen Bausteine  (Tuner, 
Soundbaustein, Speicher – EEPROM und ATMEGA168) anzusprechen . 

 

Schicht Objekt Funktion 

1 Status- und Datenregister 

des I2C-Prozessors        

PCF 8584-P 

I2C-Prozessor über NKC-IO-

Adresse ansprechen und Re-

gister schreiben oder lesen 

2 I2C-Bus  I2C-Prozessor initialisieren, n 

Datenbytes auf den I2C-Bus 

schreiben oder von ihm lesen 

3 I2C-Bausteine (Tuner, 

Soundbaustein, Speicher – 

EEPROM, ATMEGA168 als 

Slave für RDS - noch nicht 

implementiert)  

Programmierung der I2C-

Bausteine 

6.2 Schicht 1 - Beispielcode: 
 

Bemerkungen:    

 

1) Es wurden die NKC IO-Adressen $FFFFFF52  (Datenregister) und $FFFFFF53 (Statusregister) 
für den I2C-Prozessor  gewählt. 

 

 

;************************************************** ******* 

;*                                                       * 

;*       Tunertestprogramm by sEn (Sascha Neuschl)           * 

;*       Version 1.0 - 02.07.2013                        * 

;*                                                       * 

;************************************************** ******* 

;* Hardwarekonfiguration:                                * 

;*                                                       * 

;* Die Tunerkarte stellt in erster Linie ein             * 

;* Interface zum I2C-Bus von Philips dar.                * 

;* Der Interfacebaustein ist der PCF 8584, der           * 

;* über den parallelen 8-Bit-Datenbus angesteuert w ird.  * 

;************************************************** ******* 

;* Der I2C-Bus:                                          * 

;*                                                       * 

;* besteht aus dem I2C-Prozessor PCF 8584 (Master),  dem  * 

;* Tuner FM1216 ME / I H-3 dem Vorverstärker- und K lang- * 

;* stell-IC TDA 8425 und einem seriellen EEPROM AT2 4C02B * 

;* für die Speicherung von Programmplätzen für Radi o/TV  * 

;* sowie den Initialeinstellungen des Tuners und Vo rver- * 

;*stärker-/ Klangregel-ICs     * 

;************************************************** ******* 

;* Die Adressierung der Karte:                           * 

;*                                                       * 

;* liegt zurzeit auf $FFFFFF52.                          * 

;* Zur  Ansteuerung des I2C-Prozessors werden 2 Adr essen * 

;* benötigt.                                             * 

;************************************************** ******* 

;***Schicht 1: 

 

;* I2C-Basisroutinen zur Ansteuerung des I2C-Prozes sors PCF 8584 

 

;* Vereinbarungsbaustein 

 

;* Register des I2C-Prozessors PCF8584 

 

CTRLSTAT equ $fff53 ;Control-/Statusregister 

REG      equ $fff52 ;Standardregister 

 

;* I2C-Adressen der Peripheriebausteine 

 

TUNERW          equ $C0 ;tuner - write 

TUNERR          equ $C1 ;tuner - read 

TUNERIFW        equ $86 ;tuner-if - write 

TUNERIFR        equ $87 ;tuner-if - read 

 

SOUNDW  equ $82 ;sound - write 

SOUNDR  equ $83 ;sound - read 

 

MEMORW  equ $A0 ;memory - write 

MEMORR  equ $A1 ;memory - read 

 

;* Datenpufer für zu sendende oder lesende Bytes vo m I2C-Bus und nostop-Flag für 

;* senden ohne Stopcondition 

 

sendbuff:       ds.b 10 

readbuff:       ds.b 10 

nostop:         ds.b 2 

 

;* Routine sendctrl schreibt ein Byte in das Control-/Stusregister 

;* Übergaberegister ist D1 

 

sendctrl: 

move.b d1,(CTRLSTAT) 

rts 

;* Routine sendreg schreibt ein Byte in ein zuvor über das Control-/ 

;* Statusregister ausgewähltes Register. Übergaberg ister ist D2. 

 

sendreg: 

move.b d2, (REG) 

rts 

 

;* Routine readctrl liest ein Byte aus dem Control-/Statusregister 

;* Übergabergister ist D1. 

 

readctrl: 

move.b (CTRLSTAT), d1 

rts 

 

;* Routine readreg liest ein Byte aus einem zuvor über das Control-/ 

;* Statusregister ausgewählten Register. Übergaberg ister ist D2. 

 

readreg: 

move.b (REG), d2 

rts 

 

6.3 Schicht 2 - Beispielcode: 
 

Bemerkungen:    

 

1) In der Routine zur Initialisierung wurde von einer CPU-Taktrate des NKC  von 8 MHz und 2 Wait-
States  ausgegangen.  

2) Für Diagramme zum Ablauf der Initialisierung des I2C-Prozessors und des Sendens und 
Empfangens von Datenbytes an bzw. vom I2C-Bus siehe Dokument „PCF8584.pdf“. 

 

 

;*** Schicht 2: 

 

 

;* Routine INIT initialisiert den I2C-Prozessor PCF 8584 

 

init: 

move.b #$80, d1 ;Register S0' wird ausgewählt. 

jsr sendctrl 

move.b #$55, d2 ;$55 wird in Register S0' (eigene I 2C-Adresse) geschrieben. 

jsr sendreg     ;Effektive I2C-Busadresse ist damit  $AA (siehe Datenblatt) 

                ;Die Adresse ist unerheblich, weil der I2C-Prozessor nur als 

                ;Master eingestzt wird und von kein em anderen I2C-Device ange- 

                ;sprochen wird! 

                ;Aber der Schritt ist notendig, dam it der I2C-Prozessor 

                ;erkennt, ob es sich um einen INTEL - oder Motorola-Bus handelt! 

move.b #$A0, d1 ;Register S2 (Clock-Register) wird ausgewählt 

jsr sendctrl 

move.b #$18, d2 ;$18 wird in Register S2 (Clockregi ster) geschrieben. 

jsr sendreg     ;Es wird 8 MHz Systemtakt und 90 KH z I2C-SDL-Takt gesetzt. 

move.b #$C1, d1 ;Im Statusregister S1 wird das seri elle Interface des I2C- 

jsr sendctrl    ; Busses aktiviert. 

rts             ; Ende der Initialisierung 

 

 

;* Routine Master Transmitter Mode sendet n Bytes über das serielle Interface 

;* des I2C-Busseses. In D3 wird die I2C-Adresse des  anzusprechenden Peripherie- 

;* bausteins übergeben. In D4 wird die Anzahl zu se ndender Bytes übergeben, in 

;* D5 die Anzahl der gesendeten Bytes gezählt. Die zu übertragenden Bytes 

;* stehen im Sendbuffer sendbuff. 

;* Wenn keine Stopcondition ausgelöst werden soll, sondern eine Repeatedstart- 

;* condition, dann steht dies in nostop im ersten B yte als Wert $EE. 

;* Soll der Busbusy-Check übersprungen werden, so s teht dies in nostop im 

;* zweiten Byte als Wert $BB. 

 

mastertr: 

;* Initialisierung 

clr.b d5        ;Zähler für gesendete Bytes initial isieren 

lea sendbuff, A0;Sendbuffer holen 

lea nostop, A1  ;Sonderbedingungen laden für Repeat edstart oder no Busbusy-Check 

move.b 1(A1), d0;Test, ob Busbusy-Check und Startco ndition ausgeschaltet sind 

 

cmp.b #$BB, d0 

beq.s adronly 

busbusyt: 

jsr readctrl    ;Statusregister S1 auslesen 

and.b #1, d1    ;Test auf Bit BB - bus busy 

beq.s busbusyt  ;wenn 0, dann warten und nochmal le sen 

move.b d3, d2   ;I2C-Adresse des Peripheriebaustein s übergeben (Slave-Adresse) 

jsr sendreg     ;Ausgabe auf das S0-Register - Date nregister zumm Senden/ 

                ;Empfangen von Bytes des I2C-Busses  

move.b #$C5, d1 ;Startcondition für I2C-Bus wird au sgelöst                      

jsr sendctrl 

bra.s finisht   ;Mit Startcondition weiter bei fini sht 

adronly: 

move.b d3, d2   ;I2C-Adresse des Peripheriebaustein s übergeben (Slave-Adresse) 

jsr sendreg     ;Ausgabe auf das S0-Register - Date nregister zumm Senden/ 

                ;Empfangen von Bytes des I2C-Busses  

finisht: 

jsr readctrl    ;Statusregister S1 auslesen 

move.b d1, d0   ;Wert aus d1 für zweiten Bittest re tten 

and.b #128, d1  ;Test auf PIN-Bit - Übertragung bee ndet? 

bne.s finisht    ;Wenn 1, dann warten und nochmal S tatusregister lesen 

and.b #8, d0    ;Test auf LRB-Bit - slave acknowled ged? 

bne.s stopt     ;Wenn 1, dann Übertragung beenden 

cmp.b d4,d5     ;Alle Bytes übertragen? 

beq.s stopt     ;Wenn ja, dann Übertragung beenden 

addq #1, d5     ;Zähler für übertragene Bytes um 1 inkrementieren 

move.b (A0)+, d2;nächstes Byte aus Sendbuffer holen  

jsr sendreg     ;umd übertragen 

bra.s finisht   ;Schleife, solange noch nicht alle Bytes gesendet sind 

stopt: 

move.b 0(A1), d0; Test, ob Repeatedstartcondition v orliegt 

cmp.b #$EE, d0 

beq.s repstart 

move.b #$C3, d1 ;Stopcondition über Controlregister  auslösen 

jsr sendctrl 

bra.s readyt 

repstart: 

move.b #$45, d1 ;Repeatedstartconditon wird ausgelö st 

jsr sendctrl 

readyt: 

move.b #$00, 0(A1)       ;Stopcondition wieder eins chalten 

move.b #$00, 1(A1)       ;Busbusy-Check und Startco ndition wieder einschalten 

move.l #1, d0   ;Warten, bevor nächster Aufruf von mastertr oder masterec, 

jsr @delay      ;sonst zu schnell! 

rts 

 

 

;* Routine Master Receiver Mode empfängt n Bytes über das serielle Interface 

;* des I2C-Busseses. In D3 wird die I2C-Adresse des  anzusprechenden Peripherie- 

;* bausteins übergeben. In D4 wird die Anzahl zu em pfangender Bytes übergeben, 

;* in D5 die Anzahl der empfangenen Bytes gezählt. Die zu empfangenden Bytes 

;* stehen im Readbuffer readbuff. 

;* Wenn keine Stopcondition ausgelöst werden soll, sondern eine Repeatedstart- 

;* condition, dann steht dies in nostop im ersten B yte als Wert $EE. 

;* Soll der Busbusy-Check übersprungen werden, so s teht dies in nostop im 

;* zweiten Byte als Wert $BB. 

 

masterec: 

;* Initialisierung 

clr.b d5        ;Zähler für empfangene Bytes initia lisieren 

subq #1, d4     ;Zähler für zu empfangende Bytes wi rd um 1 dekrementiert, weil 

                ;das letzte Byte besonders behandel t wird 

lea readbuff,A0 ;Readbuffer holen 

lea nostop, A1  ;Sonderbedingungen laden für Repeat edstart oder no Busbusy-Check 

 

move.b d3, d2   ;I2C-Adresse des Peripheriebaustein s übergeben (Slave-Adresse) 

jsr sendreg     ;Ausgabe auf das S0-Register - Date nregister zumm Senden/ 

                ;Empfangen von Bytes des I2C-Busses  

move.b 1(A1), d0;Test, ob Busbusy-Check ausgeschalt et ist 

cmp.b #$BB, d0 

beq.s finishr 

busbusyr: 

jsr readctrl    ;Statusregister S1 auslesen 

and.b #1, d1    ;Test auf Bit BB - bus busy 

beq.s busbusyr  ;wenn 0, dann warten und nochmal le sen 

move.b #$C5, d1 ;Startcondition für I2C-Bus wird au sgelöst                     

jsr sendctrl 

finishr: 

jsr readctrl    ;Statusregister S1 auslesen 

move.b d1, d0   ;Wert aus d1 für zweiten Bittest re tten 

and.b #128, d1  ;Test auf PIN-Bit - Übertragung bee ndet? 

bne.s finishr   ;Wenn 1, dann warten und nochmal St atusregister lesen 

and.b #8, d0    ;Test auf LRB-Bit - slave acknowled ged? 

bne.s stopr     ;Wenn 1, dann Übertragung beenden 

cmp.b d4,d5     ;Alle Bytes übertragen, bis auf let ztes? 

beq.s lastbyte  ;Wenn ja, dann letztes Byte behande ln 

addq #1, d5     ;Zähler für übertragene Bytes um 1 inkrementieren 

jsr readreg     ;nächstes Byte emfangen 

move.b d2,(A0)+ ;nächstes Byte in Readbuffer schrei ben 

bra.s finishr   ; Schleife, bis vorletztes Byte emp fangen wurde 

lastbyte: 

move.b #$40, d1 ;negatives Acknowledgement wird in Controlregister gesetzt 

jsr sendctrl 

jsr readreg     ;Dies ist ein Dummyread der Slave-A dresse und wird nicht im 

                ;Readbuffer gespeichert 

finishlb: 

jsr readctrl    ;Statusregister S1 auslesen 

and.b #128, d1  ;Test auf PIN-Bit - Übertragung bee ndet? 

bne.s finishlb  ;Wenn 1, dann warten und Stausregis ter erneut auslesen 

stopr: 

move.b #$C3, d1 ;Stopcondition über Controlregister  auslösen 

jsr sendctrl 

jsr readreg     ;letztes zu empfangendes Byte holen  

move.b d2, (A0) ;und im Readbuffer speichern 

move.b #$00, 0(A1)      ;Stopcondition wieder einsc halten 

move.b #$00, 1(A1)      ;Busbusy-Check und Startcon dition wieder einschalten 

move.l #1, d0   ;Warten, bevor n chster Aufruf von mastertr oder masterec, 

jsr @delay      ;sonst zu schnell! 

rts 

 

6.4 Schicht 3 - Beispielcode: 
 

Bemerkungen:    

 

1) Am Anfang befindet sich jeweils ein Testprogramm für das EEPROM, den Soundbaustein  und 
den Tuner . 

2) Das Programm „Start“  ruft nacheinander „Initialisierung des I2C-Prozessors“ , „Einstellen des 
Soundbausteins“  und „Einstellen eines festen UKW-Senders im Tuner“  auf, sodass die 
Tunerkarte nun einen Output liefern sollte. Das EEPROM wird hier nicht verwendet. 
 
 
 
;*** Schicht 3: 
 
 
EEPROM:  ;EEPROM-Test 
 
jsr init        ;PCF8584 initialisieren 
 
;schreiben 
clr.l sendbuff          ;alle Buffer - mindestens a m Anfang - initialisieren 
clr.l readbuff 
clr.w nostop 
move.b #MEMORW, d3      ;I2C-Adresse zum Schreiben des EEPROMS 
lea sendbuff, A0        ;Sendbuffer holen 
lea nostop, A1          ;nostop-Flag holen 
move.b #$00, (A1)+      ;mit Stopcondition 
move.b #$00, (A1)       ;mit Busbusy-Check 
move.b #2, d4           ;Anzahl zu sendender Bytes 
move.b #$00, (A0)+      ;Es wird auf Adresse $00 im  EEPROM geschrieben 
move.b #$CC, (A0)       ;Es wird der Wert $CC gesch rieben 
jsr mastertr            ;und senden 
 
;lesen 
 
move.b #MEMORW, d3      ;I2C-Adresse zum  Schreiben  des EEPROMS 
                        ;Dummywrite mit Speicheradr esse des EEPROMS 

lea sendbuff, A0        ;Sendbuffer holen 
lea nostop, A1          ;nostop-Flag holen 
move.b #$EE, (A1)+      ;mit Repeatedstartcondition  
move.b #$00, (A1)       ;mit Busbusy-Check 
move.b #1, d4           ;Anzahl zu sendender Bytes 
move.b #$00, (A0)       ;Es wird im EEPROM die Adre sse $00 ausgewählt 
jsr mastertr 
 
read: 
move.b #MEMORR, d3      ;I2C-Adresse zum lesen des EEPROMS 
                        ;Jetzt Auslesen des Werts m it Leseroutine auf I2C-Bus 
lea nostop, A1          ;nostop-Flag holen 
move.b #$00, (A1)+      ;mit Stopcondition 
move.b #$BB, (A1)       ;ohne Busbusy-Check 
move.b #1, d4           ;Anzahl zu empfangender Byt es 
jsr masterec            ;und Lesen - der Readbuffer  wird durch die Routine 
                        ;masterec gefüllt! 
rts 
 
 
SOUND:  ;Sound-Test 
 
clr.l sendbuff  ;alle Buffer - mindestens am Anfang  - initialisieren 
clr.l readbuff 
clr.w nostop 
 
move.b #SOUNDW, d3      ;I2C-Adresse, um auf den So undbaustein zu schreiben 
                        ;Diesen Baustein kann man n ur schreiben! 
lea sendbuff, A0        ;Sendbuffer holen 
lea nostop, A1          ;nostop-Flag holen 
move.b #$EE, (A1)+      ;mit Repeatedstartcondition  
move.b #$00, (A1)       ;mit Busbusy-Check, weil er stes Byte 
move.b #2, d4           ;Anzahl zu sendender Bytes 
move.b #$00, (A0)+      ;Register VL 
move.b #$F8, (A0)       ;Wert 
jsr mastertr 
 
move.b #SOUNDW, d3 
lea sendbuff, A0 
lea nostop, A1 
move.b #$EE, (A1)+      ;mit Repeatedstartcondition  
move.b #$BB, (A1)       ;ohne Busbusy-Check 
move.b #2, d4 
move.b #$01, (A0)+      ;Register VR 
move.b #$F8, (A0)       ;Wert 
jsr mastertr 
 
move.b #SOUNDW, d3 
lea sendbuff, A0 
lea nostop, A1 
move.b #$EE, (A1)+      ;mit Repeatedstartcondition  
move.b #$BB, (A1)       ;ohne Busbusy-Check 
move.b #2, d4 
move.b #$02, (A0)+      ;Register BA 
move.b #$FF, (A0)       ;Wert 
jsr mastertr 
 
move.b #SOUNDW, d3 
lea sendbuff, A0 
lea nostop, A1 
move.b #$EE, (A1)+      ;mit Repeatedstartcondition  
move.b #$BB, (A1)       ;ohne Busbusy-Check 
move.b #2, d4 
move.b #$03, (A0)+      ;Register TR 
move.b #$FF, (A0)       ;Wert 
jsr mastertr 

move.b #SOUNDW, d3 
lea sendbuff, A0 
lea nostop, A1 
move.b #$00, (A1)+      ;mit Stopcondition, weil le tztes Byte 
move.b #$BB, (A1)       ;ohne Busbusy-Check 
move.b #2, d4 
move.b #$08, (A0)+      ;Register SW (Eingang 1 ode r 2) 
move.b #$CE, (A0)       ;Wert =  Eingang 1 - Tuner,  lineares Stereo 
jsr mastertr 
rts 
 
 
TUNER:  ;Tuner-Test 
 
clr.l sendbuff  ;alle Buffer - mindestens am Anfang  - initialisieren 
clr.l readbuff 
clr.w nostop 
 
;* Erst TV-Mode setzen, damit Tuningspannung nicht auf 0 stehen bleibt 
;* low band - 150 MHz 
 
move.b #TUNERW, d3      ;I2C-Adresse für Schreiben der Tunersektion 
lea sendbuff, A0        ;Sendbuffer holen 
lea nostop, A1          ;nostop-Flag holen 
move.b #$00, (A1)+      ;mit Stopcondition 
move.b #$00, (A1)       ;mit Busbusy-Check, weil er stes Byte 
move.b #5, d4           ;Anzahl zu übertragender By tes 
move.b #$0B, (A0)+      ;Dividerbyte 1 
move.b #$F5, (A0)+      ;Dividerbyte 2 
move.b #$86, (A0)+      ;Controlbyte 
move.b #$44, (A0)+      ;Bandswitchbyte 
move.b #$20, (A0)       ;Auxiliarybyte 
jsr mastertr            ;und senden 
 
;* Erst danach FM-Mode setzen und Frequnez einstell en 
 
;* FM-Mode - if part 
 
move.b #TUNERIFW, d3    ;I2C-Adresse zum Schreiben der Tuner-IF-Sektion 
lea sendbuff, A0 
lea nostop, A1 
move.b #$00, (A1)+      ;mit Stopcondition 
move.b #$00, (A1)       ;mit Busbusy-Check, weil er stes Byte 
move.b #4, d4 
move.b #$00, (A0)+      ;Erste Registeradresse, wei tere 2 durch Autoinkrement 
move.b #$4E, (A0)+      ;Switchingbyte (high sensit ivity) 
move.b #$D0, (A0)+      ;Adjustbyte 
move.b #$77, (A0)+      ;Databyte 
jsr mastertr 
 
;* Frequenz einstellen - tuner part 
 
move.b #TUNERW, d3      ;I2C-Adresse zum Schreiben der Tuner-Sektion 
lea sendbuff, A0 
lea nostop, A1 
move.b #$00, (A1)+      ;mit Stopcondition 
move.b #$00, (A1)       ;mit Busbusy-Check, weil er stes Byte 
move.b #5, d4 
                        ;eingestellte Frequnez ist 91,2 MHz 
                        ;zur Berechnung der Divider bytes siehe Datenblatt 
                        ;des Tuners 
move.b #$07, (A0)+      ;Dividerbyte 1 
move.b #$F6, (A0)+      ;Dividerbyte 2 
move.b #$80, (A0)+      ;Controlbyte 
move.b #$19, (A0)+      ;Bandswitchbyte 

move.b #$B0, (A0)       ;Auxiliarybyte 
jsr mastertr 
rts 
 
 

start:  ;Gesamttest der Karte 
 
jsr init        ;I2C-Prozessor initialisieren 
jsr sound       ;Soundbaustein einstellen 
jsr tuner       ;Tuner einstellen 
rts             ; ...und nun hört man endlich Radio  am NKC :-) 
 
end 

7 RDS-Dekoder 

7.1 Ansatz 
Da es sehr viele Projekte für einen RDS-Dekoder mit einem ATMEGA-Prozessor im Internet gibt, 
wollte ich an dieser Stelle das Rad nicht neu erfinden. Allerdings handelte es sich fast durchweg um 
Lösungen, die mit einem LCD-Display arbeiten, und es ergab sich die Frage, wie ich die Daten in den 
NKC kriegen sollte.  

Die einfachste Methode schien ein Transfer über die RS 232-Schnittstelle  zu sein, so denn der 
ATMEGA und der NKC schnell genug wären und in der Kommunikation zusammenpassen würden. 

Marc verwendet seine Lösung in Zusammenhang mit einer Hauppauge-Karte in seinem PC und lässt 
sich die RDS-Texte via Hyperterminal auf seinem PC anzeigen. Genau das hatte ich gesucht. Vielen 
Dank an Marc an dieser Stelle! 

Ich fand das folgende Projekt „Atmel AVR Atmega168 RDS decoder with serial output“  von 
Marc Ketel : http://atoomnet.net/atmel-avr-atmega168-rds-decoder-with-serial-output/                     
Dort gibt es eine Beschreibung und auch den Code für den ATMEGA 168-P. 

Ich änderte die Übertragungsgeschwindigkeit für die RS 232-Schnitts telle  im Code des ATMEGA 
168-P von „19.200“  auf „9.600 BAUD“ , weil der NKC mit einer höheren Geschwindigkeit in meinem 
System (68008 mit 8 MHz) nicht klarkommt.  

Durch diese geringere Übertagungsgeschwindigkeit ergibt sich ein Laufzeitproblem  bei der Darstel-
lung des RDS-Texts (RTA/RTB) . Die ersten 3 Zeichen werden fehlerhaft an den NKC übertragen.  

Deshalb habe ich den Original-Code so geändert, dass der RDS-Text nur jedes zweite Mal 
ausgegeben  wird. Jedes erste Mal wird eine Stringvariable mit dem RDS-Text gefüllt , um ihn mit 
dem neu zu empfangenden RDS-Text zu vergleichen , um die Mehrfachausgabe zu 
unterdrücken . Der Stringvergleich  u n d  die Ausgabe zusammen dauern zu lange. Da die Aussen-
dung des RDS-Texts mehrfach mit demselben Inhalt geschieht, findet kein Informationsverlust statt. 

Die Änderungen habe ich im Code im Abschnitt 7.7 in Rot hervorgehoben. 

7.2 Aktueller Stand 
Thomas Heller und Matthias Kübeck haben nun eine weitere Möglichkeit geschaffen, RDS-Zeichen an 
den NKC zu übertragen. Dazu wird der ATMEGA 168-P  (der die Signalaufbereitung des RDS-Daten-
stroms vornimmt) als Slave an den I2C-Bus  angebunden. Über den I2C-Prozessor PCF 8584-P  
werden die RDS-Zeichen aus dem ATMEGA 168-P ausgelesen  bzw. der RDS-Dekoder gesteuert. 

Es gibt also aktuell zwei Methoden, den RDS-Dekoder an den NKC anzubinden: 

 

Methode Standard Schnittstelle 

1 optional 

(Debug) 

RS232:                        

Adresse = Standardadresse, wie sie das Grund-programm für die SER- / 
SER2-Karte vorsieht 

Konfiguration = 9.600 BAUD, 8 Bit, keine Parität, 1 Stoppbit (Dies kann 
im Code des ATMEGA 168-P und im Programmcode der Tunerkarte 
geändert werden. Allerdings ist dies die höchste Baudrate, die ich mit 
einer 68008 CPU bei 8 MHz Prozessortakt verarbeiten konnte.) 

2 ja I2C-Bus (noch nicht implementiert) 

7.3 Schaltungsprinzip: 
 

R D S-

Demodulator

TDA3730B

Microcontroller

ATMEGA 168

MPX-Signal

Mit

R D S-Daten

N
K

C
6

8
0

X
X

Z
8

0

R D S

Serielle Daten

(9.600, 8, N, 1)

RDS-Bitstrom

HF-Radio

Tuner

Philips 

FM1216

ME / I H-3

LED

group

complete

LED

heartbeat

LED
synchron

quality
SER/SER2

GDP

KEY

Anzeige

Tastatur-

zeichen

I2C-Prozessor

PCF8584

Datenbuspuffer

74HCT245

R D S

Parallele Daten

(D0 – D7)

R D S

Parallele Daten

(D0 – D7)

SDA

SCL

 

 

 

Der Philips-Tuner FM1216 ME / I H-3  liefert an PIN 14 das MPX-Signal , aus dem die RDS-Daten  
gewonnen werden.  

Der Demodulatorbaustein TDA 3730B  filtert aus dem MPX-Signal  einen Bit-Strom  aus und liefert 
ihn an den Microcontroller ATMEGA 168-P .  

Dieser dekodiert aus dem Bitstrom  zunächst „Zeichen“  und interpretiert aus den Zeichen  die 
relevanten RDS-Strukturen mit ihren Gruppen und Informationen  (siehe z.B. 
http://de.wikipedia.org/wiki/Radio_Data_System oder Datenblatt „RDS-GRUNDLAGEN.pdf“). 

 

 

RDS-Gruppe aus 4 Blöcken: 

 

 

 

Beispielbitmuster für die Gruppe „11A“: 

Die ausgewerteten Informationen  werden vom ATMEGA 168-P  im Standard  über den I2C-Bus  
bereitgestellt. Optional  werden die Informationen  über die RS 232-Schnittstelle  des Microcon-
trollers ausgegeben und können einer SER- / SER2-Karte  des NKC zugeführt werden. 

Es können nicht nur RDS-Informationen von dem ATMEGA 168-P empfangen, sondern auch Zeichen 
zur Programmsteuerung übergeben  werden. Im Programm des Microcontrollers sind folgende Zei-
chen mit Funktionen definiert: 

 

 

Taste Funktion 

„r“ Reset des Dekoders  

„G“ bzw. „g“ Gruppenanzeige „ein“ / “aus“  

„B“ bzw. „b“ Anzeige Bitstrom „ein“ / „aus“  

 

 

Auf der Platine befinden sich drei LEDs , die eine Aussage über den RDS-Datenstrom  machen und 
eine weitere LED , die den Interrupt-Betrieb des I2C-Prozessors PCF 8584-P  anzeigt: 

 

 

LED-Farbe Funktion 

Rot Heartbeat bzw. Bitstrom vorhanden  

Grün RDS-Gruppe vollständig empfangen  

Gelb Synchronqualität  

Blau Interrupt-Betrieb PCF 8584-P 

7.4 Funktionen des RDS-Dekoder-Programms: 
 

Die Software des Microcontrollers kann folgende RDS-Informationen ermitteln: 

 

Nr. Funktion 

1 Program Identification code (PI: 0x83C7, Detected new station) 

2 Program service name (PS: RADIO538) 

3 Program Type code (PTY: 0x0A, Pop Music.) 

4 Traffic Program Identification code & Traffic announcement code (TP&TA: 0×01 
0×00, Traffic announcements available on this station and maybe via EON on 
another station.) 

5 Music Speech switch code (MS: 0×01, Music is being broadcasted or station does 
not use MS flag.) 

6 Decoder-identification control code (DI: 0×01, Stereo, Dynamic PTY.) 

7  Alternative frequency codes (AF: 0×98, 102.7MHz.) 

8  Linkage Actuator (LA: 0×00) 

9  Extended Country Code (ECC: 0xE3) 

10  Radio Text (RTA: Radio 538 = Randstad (Zuid) 102.7 FM) 

11 Clock-time and date (CT: 0x1A53CD844, UTC 2006-07-02 (MJD 53918) 13:33:00 
+02:00, TIME 15:33:00) 

RDS-Zeichenstrom: 

 

 

 

Unkodierte Gruppenanzeige: 

7.5 Kommunikation zwischen RDS-Dekoder und NKC 
Die Kommunikation des ATMEGA 168-P  über den I2C-Bus  oder die RS232-Schnittstelle  gestaltet 
sich recht problemlos 

Bei Betrieb mit der RS232-Schnittstelle  sind die korrekten Einstellungen  im Programm des Micro-
controllers und für die SER- / SER2-Kartede vorzunehmen. 

Wie man auf dem Bild der RDS-Zeichnestroms  sieht, werden Umlaute nicht standardmäßig vom 
NKC dargestellt .  Es sollte ja „Antenne Münster“ heißen. Hier muss man die ankommenden 
Zeichencodes für Umlaute im Programm auf dem NKC von z.B. „Ü“ in „UE“patchen. 

Für die Anzeige einer „stehenden Zeile“ , wie man sie aus Autoradios kennt, muss man sich die rele-
vante Info aus dem Text „ausschneiden“  und in eine Variable speichern. Das Programm in dem 
ATMEGA 168-P schreibt die empfangenen Daten einfach nacheinander weg. 

 

7.6 Test des RDS-Dekoders 
Ein Test des RDS-Dekoders  lässt sich in zwei Schritten durchführen. Zuerst wird ein Sender im 
Tuner eingestellt und das entsprechende Programm dafür verlassen. Der Tuner läuft mit seinen Ein-
stellungen ja weiter. Dann wird ein „Terminalprogramm“  gestartet, das zusammen mit einer seriel-
len Schnittstellenkarte  das Empfangen und Senden von Zeichen ermöglicht.  

Natürlich kann auch mit dem Bedienprogramm aus Abschnitt 9 getestet werden. 

 

7.7 Testpunkte auf der Platine 
 

Testpunkt Beschreibung 

T1 TDA 7330B – PIN 4: FILOUT (Filterausgang) 

T2 TDA 7330B – PIN 11: T57 (Taktausgang mit 57 KHz) 

T3 TDA 7330B – PIN 12: RDCL (RDS Takt) 

T4 TDA 7330B – PIN 13: RDDA (RDS Daten) 

7.8 Schaltplan 

7.9 Programmierung des Microcontrollers ATMEGA 168- P 
für den RDS-Dekoder 

 

Das ursprüngliche Programm vom Marc Ketel wurde in 2 Punkten geändert: 

- Die Baudrate der RS232-Schnittstelle wurde auf 9.600 Baud geändert. 
- Der RDS-Text wird nur jedes 2. Mal ausgegeben, um Timing-Probleme mit dem NKC zu 

vermeiden. 

 

7.9.1 Programmkopf und Verweis auf ATMEL Studio Pro jekt 
/* 
 
Copyright (C) 2006 Marc Ketel 
 
This program is free software; you can redistribute  it and/or 
modify it under the terms of the GNU General Public  License 
as published by the Free Software Foundation; eithe r version 2 
of the License, or (at your option) any later versi on. 
 
This program is distributed in the hope that it wil l be useful, 
but WITHOUT ANY WARRANTY; without even the implied warranty of 
MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE .  See the 
GNU General Public License for more details. 
 
You should have received a copy of the GNU General Public License 
along with this program; if not, write to the Free Software 
Foundation, Inc., 51 Franklin Street, Fifth Floor, Boston, MA  02110-1301, USA. 
 
#define INFO "20060629-1 RDS Decoder by Marc Ketel alias DiNo, marc@atoomnet.net." 
 
Works with RDS clock signal and RDS data signal. Du ring design a TDA7300B rds 
demodulator was used. 
 
This source compiles correctly with WinAVR 20060421 . 
Use a Atmega168 to flash program into. low fuse: 0x F0, high fuse: 0xDD 
 
Versions: 
 
  20060629-1  Initial version 
   
  20180127-2  Version for use with 9.600 baud serial transfer by Sascha Neuschl sEn 

 With initial version the serial transfer of RDS te xt lasts too long   
     and leads to errors within the first 3 to 4 ch aracters.  

 To fix this RDS text is printed just every second time, not every     
 time! A print flag p is introduced. 

 

Das Programm gibt es als ATMEL Studio Projekt: 

8 S-Meter - Messung der Signalstärke 

8.1 Ansatz 
Der Philips FM1216 ME / I H-3 bietet datentechnisch keinen Wert über die Signalstärke des 
empfangenen Senders an. Man kann dort nichts über den I2C-Bus abfragen. 

Als Ersatz für die Messung der tatsächlichen Signalstärke eines Senders am Antenneneingang, wird 
in vielen Empfangsgeräten die Tuner-Regelspannung (AGC) herangezogen. Dies ist eine Spannung, 
die in der IF section des Tuners erzeugt und der Tuner section zugeführt wird, um eine Übersteuerung 
des Tuners durch einen starken, örtlichen Sender zu verhindern und dafür zu sorgen, dass das IF-
Signal immer möglichst konstant hoch ist. 

Dieses AGC-Signal verhält sich nun so, dass es bei diesem Tuner bei einem schwachen Sender 
hoch  ist und bei einem starken niedrig . 

 

8.2 Schaltungsprinzip: 
 

 

 

Der Philips FM1216 ME / I H-3 liefert die Tuner-Regelspannung an PIN 1. Diese darf nur sehr hoch-
ohmig – mit minimal 10 Megaohm  – belastet werden, sonst steigt der Tuner aus.  

Deshalb wird sie mit dem Operationsverstärker µA 741  als nicht-invertierender Verstärker  
getrieben. Die Verstärkung ist auf ca. 1,8-fach eingestellt (10 Kilohm / 5,6 Kiloohm). Sie kann mit dem 
Trimmer R11 justiert werden. Eine Z-Diode verhindert, dass die Spannung größer als 5 Volt werden 
kann. Diese Spannung kann jetzt mit einem Analog-Digital-Wandler ausgewertet werden. Ich nutze 

dazu eine AD10X1-Karte . In einem weiteren Ausbau soll der ADC des ATMEGA168 verwendet 
werden. 

Die Treiberschaltung ist in der unteren, rechten Ecke des Tunerschaltplans auf Seite 9 und hier als 
Ausschnitt angegeben.  

 

8.3 Schaltplan - Ausschnitt: 

8.4 Kalibrierung: 
Die Signalmessung muss kalibriert werden, damit eine sinnvolle Anzeige der Signalstärke erfolgen 
kann. Dazu wird einerseits die maximale Spannung am Ausgang des Operationsverstärkers µA 741 
mit dem Trimmer R11 eingestellt und des Weiteren die Auswertung des Signalwerts des ADCs im 
Bedienprogramm angepasst. 

 

8.4.1 Ausgangspannung des Operationsverstärkers: 
Zunächst wird die UKW-Antennenbuchse kurzgeschlossen, sodass der Tuner kein Signal 
empfängt. Jetzt kann die höchste Spannung mit R11  eingestellt werden, die zu erwarten ist, 
da die AGC-Spannung bei einem schwachen Sender am höchsten ist. 

Ich habe diese Spannung auf 4 Volt  eingestellt. 

Jetzt wird die Antenne an die UKW-Antennenbuchse angeschlossen. Z.B. mit der „Auto-Suche“ 
eines Senders im Bedienprogramm schreite man nun das gesamte Frequenzband ab und 
notiert die Spannungswerte. Mit meiner Antenne ergibt sich eine Differenz von höchster zu 
niedrigster Spannung  von 1,7 Volt . 

 

8.4.2 Anpassung des Bedienprogramms: 
Zu den oben ermittelten Spannungen kann man sich den daraus ermittelten Wert des ADC  in 
der oberen rechten Ecke der Hauptmaske des Bedienprogramms anzeigen lassen, wenn man 
dieses Code-Stück wieder einkommentiert: 

 

;* AD10X1 - Test: Für Test wird AD-Wert oben rechts  in der Hauptmaske 
angezeigt 

;lea ad10x1(pc), a0 

;jsr @print4d 

;moveq #3-1, d2 

;loop: 

;move.b #' ', (a0)+ 

;dbra d2, loop 

;clr.b (a0) 

;lea ad10x1(pc), a0 

;move #$22, d0 

;move #440, d1 

;move #190, d2 

;jsr @write 

 

Bei mir ergeben sich die Werte 384 – 213.  Der Wert des ADCs wird im Bedienprogramm 
abgefragt. Es sind 5 Signalstufen (Balken) definiert. Ich habe die Wertdifferenz 171  nun linear  
auf die 5 Balken verteilt. Die Anzeige hat also keinen technischen Hintergrund wie z.B.Dezibel. 
Die Definition der 5 Signalstufen kann je nach Bedarf geändert werden. 

;* Auswertung des gewandelten Wertes für 5 Anzeigeb alken 

;* Je nach Auswertung wird eine Bitkombination in D 3 geladen und 
darüber 

;* bestimmt, ob ein Balken mit Schreibstift oben od er unten gezeichnet 
wird. ADC-Wert in D0. 

 

;* kein Balken 

cmp.w #323, d0 

ble.s b1 

move.b #%00011111, d3 

bra.s grafik 

;* 1 Balken 

b1: 

cmp.w #296, d0 

ble.s b2 

move.b #%00011110, d3 

bra.s grafik 

;* 2 Balken 

b2: 

cmp.w #269, d0 

ble.s b3 

move.b #%00011100, d3 

bra.s grafik 

;* 3 Balken 

b3: 

cmp.w #242, d0 

ble.s b4 

move.b #%00011000, d3 

bra.s grafik 

;* 4 Balken 

b4: 

cmp.w #215, d0 

ble.s b5 

move.b #%00010000, d3 

bra.s grafik 

;* 5 Balken 

b5: 

move.b #%00000000, d3 

9 Bedienprogramm  
 

Es gibt auch ein fertiges Bedienprogramm in 680XX-Assembler für die Tuner-Karte.                                  
Die aktuelle Version ist V3.0 und als Dateien ‚nkcradio.asm‘  und ‚nkcradio.68k‘  verfügbar. 

9.1 Parameter und Systemkonfiguration 
- Bibliothekseintrag 
- Basisadresse der Tunerkarte (I2C-Prozessor PCF8584)  
- I2C-Adressen von Tuner, Soundbaustein und EEPROM  
- Serielle Schnittstellenkarte (RDS): 

o Art (SER / SER2) 
o Konfiguration 

- Analog-Digital-Wandler-Karte (Signalstärke über AGC ): 
o Fix AD10X1 

- Uhrenkarte 

 

9.2 Funktionen 
 

- Einstellung der zu empfangenden Frequenz: 
o Direkt per Eingabe einer Frequenz 
o Erhöhen oder Vermindern einer Frequenz um 100 KHz 
o Automatische Suche des nächsten Senders – nur aufwärts 
o Speichern eines Senders 
o Aufruf eines gespeicherten Senders 
o Aufruf der Programmliste und Auswahl eines Programms/Senders 

- Audioeinstellungen: 
o Lautstärke 
o Höhen 
o Bässe 

- Anzeigen: 
o Datum und Uhrzeit, wenn eine Uhrenkarte im NKC vorhanden ist 
o Stereo, wenn der Sender in Stereo empfangen wird 
o Signalstärke des empfangenen Senders in 5 Signalstufen, wenn an ST7 eine 

AD10X1-Karte angeschlossen ist 
o RDS: 

 Anzeige des RDS-Stationsnamens der gerade empfangenen Station 
 Anzeige des Radio-Textes der gerade empfangenen Station 
 Anzeige des RDS-Lauftextes in einer separaten Maske 

o Anzeige der gespeicherten Sender als Programmliste in einer separaten Maske 
o Einstellung von Lautstärke, Höhen und Bässen 
o Bedienstatus (z.B. Frequenz, Lautstärke, höhen, Bässe, Sender speichern …) 

- Speicherung von Einstellungen und Sendern auf dem E EPROM der Karte: 
o Die letzten Einstellungen für Frequenz und Audio werden auf dem ersten 

Speicherplatz im EEPROM gespeichert. Bei Neustart des Programms wird mit diesen 
Werten gearbeitet 

o Es können 15 Sender mit ihrer Frequenz und dem RDS-Stationsnamen in einer 
Programmliste gespeichert werden 

- Bildschirmschoner 

9.3 Masken 
 

Das Programm gliedert sich in die Folgenden Masken: 

- Hauptmaske: 

 

 
- Maske für RDS-Zeichenstrom: 

- Maske für gespeicherte Sender (Programmliste): 

 

 

 

10 Stückliste 

10.1 Tunerkarte 

10.1.1 Platine 
Für die Platine liegen PCB-Dateien vor: 

10.1.2 Bauteile: 

11 Anhang 

11.1 Datenblätter TTL-Bausteine:  

11.1.1 74LS688 
 

 

 

 

11.1.2 74LS245 

11.2 Datenblatt RS232-Pegelwandler:  

11.2.1 MAX232 

11.3 Datenblatt Operationsverstärker:  

11.3.1 µA 741-P 

11.4 Verweis auf Datenblätter komplexer Bausteine u nd 
Spezifikationen 

 

Baustein/Objekt Funktion Datenblatt 

I2C-Bus I2C-Bus-Spezifikation UM10204.pdf 

PCF 8584-P I2C-Prozessor PCF8584.pdf 

Philips FM1216 

ME / I H-3 

Tuner FM1216ME_MK3.pdf 

TDA 8425 Sound TDA8425.pdf 

AT24C02 Speicher AT24C02.pdf 

TDA 3730B RDS-Demodulator TDA7330B.pdf 

ATMEGA 168-P Microcontroller ATMEGA_168.pdf 

RDS RDS - Spezifikation RDS_GRUNDLAGEN.pdf 

 




